



NATIONAL RADIO ASTRONOMY OBSERVATORY

ELECTRONICS DIVISION TECHNICAL NOTE NO. 96

TITLE: LADDER ANALYSIS PROGRAM

AUTHOR(S): Roger D. Norrod

DATE: May 28, 1981

Distribution: M. Balister
S. Weinreb
J. Payne
P. Napier
R. Lacasse
H. Hvatum
C. Burgess

NATIONAL RADIO ASTRONOMY OBSERVATORY
Green Bank, West Virginia

MEMORANDUM

May 28, 1981

To: Green Bank Electronics Division
From: Roger Norrod
Subj: LADDER Analysis Program

Introduction:

An APPLE computer system is now available at Green Bank for use by the Electronics Division. The purpose of this memo is to acquaint potential users with a program available on disk. The LADDER program was written by Sandy Weinreb and calculates the input impedance of a network. Results can be displayed in table form or plotted on a Smith chart. The program consists of a group of subroutines which perform various functions. The user writes a series of BASIC language statements describing his network and calling the proper subroutines.

To use the program, first insure that a disk marked NRAO LIBRARY 1.0 is in the disk drive. Then type RUN LADDER and return. After a few seconds a "menu" will be displayed (Figure 1). Just follow instructions on the menu to execute the program. Table 1 describes the functioning of the menu subprograms.

To describe his network, the user writes a BASIC language program in lines 100 to 998. Table 2 lists subroutines the user can call from within his program, and Table 3 gives a partial list of variables with special meaning to LADDER. To insert his own program into LADDER, the user should type LOAD LADDER, and then DEL 100,998 to eliminate any old programs. He then types in his own program and SAVE LADDER to store his version on disk.

Example 1:

Figure 2 is a listing of a program used to analyze a three-pole bandpass filter. For those not familiar with Applesoft BASIC, anything following REM (for remark) is ignored by the computer and colons are used to string statements together. Lines 140 through 210 are executed for each frequency defined in line 130. Line 150 initializes RN and XN to the terminating impedance, in this case 75 ohms. Line 160 defines the first filter element, a 2258 pF capacitor in parallel, and calls subroutine 1300. This subroutine converts RN and XN to admittance, combines it with the admittance of the capacitor, converts this total admittance to an impedance, and stores it in RN and XN. Lines 170 through 200 add elements until RN and XN have the input impedance of the total network. Line 210 calculates the resulting reflection coefficient, prints it on the screen and plots it on a Smith chart overlay. Line 220 loops the program back for the next frequency. When all frequency points have been completed, line 999 sends the computer back to wait for the

user to select another menu item. Figure 3 shows the printout caused by selecting menu items (T) and (P) after running the above program. The program takes about three seconds per frequency.

Example 2:

A program written to analyze a coaxial bandpass filter with design cutoff frequency of 10 GHz is listed in Figure 4. Subroutine 1400 is used to rotate impedances through transmission lines. Figure 5 shows the printout for this program. About five seconds were required for each frequency.

Program Listing:

A complete listing of LADDER is included with this memo. Table 4 lists some subroutines used by LADDER internally. These would normally not be called by the user program but are available if needed.

Housekeeping:

Manuals describing the use of the APPLE system are normally kept on the rack housing the computer. A three-ring binder labeled ADIOS #4 has much useful information. I will be glad to assist anyone interested in getting started.

Anyone using LADDER much would probably be wise to have their own disk so that other users don't erase their program. We have several blank disks and copying the NRAO LIBRARY 1.0 disk takes only a few minutes.

Future Expansions:

I hope to add a subroutine in the near future that will calculate the fringing capacitance of a stepped coaxial line. Anyone else who has a suggestion for a useful subroutine, let me know so that it can be made available to all users.

RDN/cjd

Enclosures:

1. Table 1 - LADDER Subprograms Accessed thru Menu
2. Table 2 - LADDER User Subroutines
3. Table 3 - LADDER Variables
4. Table 4 - LADDER Subroutines used by LADDER
5. Figure 1 - Menu Displayed by LADDER
6. Figure 2 - Analysis of Lumped Element Bandpass Filter
7. Figure 3 - Printout for Lumped Element Bandpass Filter
8. Figure 4 - Analysis of Coaxial Lowpass Filter
9. Figure 5 - Printout for Coaxial Lowpass Filter
10. Program Listing

TABLE 1

LADDER SUBPROGRAMS ACCESSED THRU MENU

<u>Menu Item</u>	<u>Line No.</u>	<u>Function</u>	<u>Returns to</u>
H	74	Clears screen.	End
L	4200	Prints list of lines 100-999.	Menu
R	84	Executes user program.	Menu
S	5014	Loads fresh Smith chart.	Menu
T	9300	Displays table of calculated data with hard copy option.	Menu
U	5000	Loads utilities (Smith chart and HGR shapes).	Menu
P	6500	Prints copy of graphics table.	Menu

TABLE 2

LADDER USER SUBROUTINES

<u>Line No.</u>	<u>Function</u>	<u>Inputs</u>	<u>Outputs</u>
1000	Admittance of Series R, L, C	R, L, C, W	G, B
1010	Impedance to Admittance	RE, X	G, B
1100	Impedance of Parallel R, L, C	R, L, C, W	RE, X
1110	Admittance to Impedance	G, B	RE, X
1200	Series Branch, Series R, L, C	R, L, C, W, RN, XN	RN, XN (new)
1300	Shunt Branch, Series R, L, C	R, L, C, W, RN, XN	Rn, XN (new)
1400	Series Xmission Line	W, LN, Z0, RN, XN	RN, XN (new)
1500	ZN to $\Gamma = SR + j SI$	RN, XN, R0	SR, SI
1600	Γ to ZN = RN + j XN	SR, SI, R0	RN, XN
2000	Complex Multiply: $C = A * B$	AR, AI, BR, BI	CR, CI
2100	Complex Reciprocal: $C = 1/B$	BR, BI	CR, CI
2200	Complex Divide: $C = A/B$	AR, AI, BR, BI	CR, CI
8000	Smith Chart Plot	SR, SI	---
9000	Maxes Table on CRT and Stores F, RN, XN, SR, SI, SWR	F, RN, XN	SR, SI

TABLE 3

LADDER VARIABLES
(Partial List)

<u>Variable</u>	<u>Description</u>
R	Resistance of Circuit Branch, Ohms
L	Inductance of Circuit Branch, nH
C	Capacitance of Circuit Branch, pF
W	Radian Frequency, $2\pi * \text{GHz}$
RE	Real Part of Impedance
X	Imaginary Part of Impedance
G	Real Part of Admittance
B	Imaginary Part of Admittance
RN	Real Part of Circuit Input Impedance
XN	Imaginary Part of Circuit Input Impedance
SR	Real Part of Reflection Coefficient
SI	Imaginary Part of Reflection Coefficient
Z \emptyset	Characteristic Impedance of Transmission Line
LN	Length of Transmission Line, mils
AR, BR, CR	Real Part of Complex Number A, B, or C
AI, BI, CI	Imaginary Part of Complex Number A, B, or C
TP	2π , defined by LADDER for user

TABLE 4

LADDER SUBROUTINES USED BY LADDER


<u>Line No.</u>	<u>Function</u>	<u>Called By</u>
5300	Turns on Printer	9200, 4200
9100	Stores Data in Arrays	9000
9200	Prints Table of Calculated Data	9300

```
LADDER
NETWORK ANALYSIS PROGRAM
NR90--APRIL 1, 1981
NORMAL UNITS ARE GHZ,MILS,NH, AND PF
UTILITIES SHOULD BE LOADED AT START OF EACH SESSION.
TYPE (H) TO HALT
(L) TO LIST MAIN PROGRAM
(R) TO RUN ANALYSIS
(S) TO LOAD FRESH SMITH CHART
(T) TO MAKE HARD COPY OF DATA
(U) TO LOAD UTILITIES FROM DISK
(P) TO PRINT GRAPHIC TABLE

(H),(L),(R),(S),(T),(U),OR (P)?
```

FIGURE 1

MENU DISPLAYED BY LADDER



```

100 REM LUMPED ELEMENT BPF
105 REM 75 OHM, 0.50B CHEBYSCHV, 3POLE, 6.0-7.5 MHZ BANDPASS
110 R0 = 75: REM DEFINE CENTER OF SMITH CHART
120 C1 = 2258:L1 = 249:C2 = 64.5:L2 = 8730: REM DEFINE ELEMENT VALUES
130 FOR F = 5.5 TO 8.0 STEP 0.1
140 W = TP * F / 1000: REM CONVERT MHZ TO GHZ
150 RN = 75: XN = 0: REM TERMINATION=75 OHMS
160 R = 0:L = 0:C = C1: GOSUB 1300: REM 1ST SECTION
170 R = 0:L = L1:C = 0: GOSUB 1300: REM 1ST SECTION COMPLETED
180 R = 0:L = L2:C = C2: GOSUB 1200: REM 2ND SECTION
190 R = 0:L = L1:C = 0: GOSUB 1300: REM 3RD SECTION
200 R = 0:L = 0:C = C1: GOSUB 1300: REM 3RD SECTION COMPLETED
210 GOSUB 9000: GOSUB 8000: REM STORE & PLOT INPUT IMPEDANCE
220 NEXT F
999 GOTO 72: REM END ANALYSIS, GO TO PROMPT

```

FIGURE 2

ANALYSIS OF LUMPED ELEMENT BANDPASS FILTER

F	REAL Z	IMAG Z	/GAMMA/	ANGLE	USHR
5.5	.6	33.2	.987	132.3	153.3
5.6	1.3	39.1	.974	124.9	76.3
5.7	3	48.1	.944	114.6	34.8
5.8	9	63.2	.869	99.3	14.28
5.9	35.1	87.2	.683	76.2	5.31
6	103.4	55.4	.333	45.6	2
6.1	70.8	-1.2	.03	-164	1.06
6.2	45.7	2.8	.244	173.1	1.65
6.3	39.4	12.1	.327	155.2	1.97
6.4	41.9	19.8	.325	139.5	1.96
6.5	51.2	24	.263	124	1.71
6.6	65.3	19.7	.155	108.1	1.37
6.7	74.8	3.1	.021	91.8	1.04
6.8	69.1	-15.6	.115	-104.4	1.26
6.9	55.6	-23.1	.228	-119.9	1.59
7	45.2	-21.3	.3	-134.5	1.86
7.1	40.4	-15.3	.325	-148.6	1.96
7.2	41.8	-7.7	.291	-163.2	1.82
7.3	51.9	0	.182	-179.9	1.44
7.4	79.1	-1.6	.028	-20.4	1.06
7.5	104.3	-53.9	.328	-44.8	1.97
7.6	49.3	-91.5	.616	-69.3	4.2
7.7	16.5	-73.9	.802	-89.4	9.09
7.8	6.5	-58.3	.898	-104	18.55
7.9	3	-48.2	.945	-114.5	35.09
8	1.6	-41.4	.968	-122.2	61.87

6.0-7.5 MHZ BPF




FIGURE 3

PRINTOUT FOR LUMPED ELEMENT BANDPASS FILTER


```

100 REM SEVEN SECTION, CHEBYSHEV, COAXIAL LPF. 0.1 DB RIPPLE.
110 R0 = 50: REM DEFINE CENTER OF SMITH CHART
120 FOR F = 1.0 TO 15.0 STEP 1
130 W = TP * F:RN = 50.0:ZN = 0.0: REM TERMINATION = 50 OHMS
140 R = 0:L = 0:C = .02: GOSUB 1300: REM FRINGE CAP
150 Z0 = 15:LN = 27: GOSUB 1400: REM 1ST SECTION
160 R = 0:L = 0:C = .04: GOSUB 1300: REM FRINGE CAP
170 Z0 = 85:LN = 170: GOSUB 1400: REM 2ND SECTION
180 R = 0:L = 0:C = .04: GOSUB 1300: REM FRINGE CAP
190 Z0 = 15:LN = 49: GOSUB 1400: REM 3RD SECTION
200 R = 0:L = 0:C = .04: GOSUB 1300: REM FRINGE CAP
210 Z0 = 85:LN = 194: GOSUB 1400: REM 4TH SECTION
220 R = 0:L = 0:C = .04: GOSUB 1300: REM FRINGE CAP
230 Z0 = 15:LN = 49: GOSUB 1400: REM 5TH SECTION
240 R = 0:L = 0:C = .04: GOSUB 1300: REM FRINGE CAP
250 Z0 = 85:LN = 170: GOSUB 1400: REM 6TH SECTION
260 R = 0:L = 0:C = .04: GOSUB 1300: REM FRINGE CAP
270 Z0 = 15:LN = 27: GOSUB 1400: REM 7TH SECTION
280 R = 0:L = 0:C = .02: GOSUB 1300: REM FRINGE CAP
290 GOSUB 9000: GOSUB 8000: REM PLOT POINT ON SMITH CHART
300 NEXT F
999 GOT0 72: REM END ANALYSIS, GO TO PROMPT

```

FIGURE 4

ANALYSIS OF COAXIAL LOWPASS FILTER

F	REAL Z	IMAG Z	/GAMMA/	ANGLE	USHR
1	49.7	-.5	6E-03	-119.6	1.01
2	50.2	.1	2E-03	30.8	1
3	53.2	.1	.031	.9	1.06
4	56.7	-4.1	.074	-29.2	1.16
5	54.9	-10.6	.111	-59.6	1.25
6	48.6	-11.5	.117	-90.3	1.26
7	46.1	-5.6	.071	-121.8	1.15
8	52.6	1.3	.028	25.2	1.06
9	66.4	-3.1	.143	-9.2	1.33
10	63.1	-17.9	.194	-44.9	1.48
11	50.1	-5.6	.056	-85.3	1.12
12	75	49.2	.411	41.6	2.4
13	269.7	-275.6	.835	-10.7	11.12
14	8	-126.4	.958	-43	46.35
15	1.2	-81.7	.987	-62.9	147.59

50 OHM LPF




FIGURE 5

PRINTOUT FOR COAXIAL LOWPASS FILTER

PROGRAM LISTING

PROGRAM LENGTH= 3040 BYTES VARIABLES= 35 BYTES
 FREE MEMORY= 12404 BYTES
 START=16385 LOMEM=24425 FREE=24460 STRING=36864 HIMEM=36864

```

10 REM                   RLADDER VERSION 21MAY81

12 REM MOVE PROGRAM TO HIGH MEMORY
14 PRINT CHR$(4);"BRUN LOMEM:" & LOMEM: 16384
16 HIMEM: 36864
18 POKE 1013,76: POKE 1014,12: POKE 1015,151
20 POKE 1016,76: POKE 1017,0: POKE 1018,151
25 PRINT CHR$(4);"BRUN LOMEM:" & LOMEM: 16384
30 TEXT : HOME :U = 2
32 UTAB U: HTAB 16: PRINT "LADDER"
33 UTAB U + 3: HTAB 8: PRINT "NETWORK ANALYSIS PROGRAM "
34 UTAB U + 6: HTAB 10: PRINT "NRAO—APRIL 1, 1981"
35 UTAB U + 10: PRINT "NORMAL UNITS ARE GHZ,MILS,NH, AND PF": PRINT
  "UTILITIES SHOULD BE LOADED AT START OF EACH SESSION.": UTAB U +
  15: REM

38 REM INITIALIZE CONSTANTS
40 D$ = CHR$(4): REM D$=CTRL D
45 IN = 1E06: DIM FS(50),RS(50),XS(50),MS(50),AS(50),US(50)
48 HCOLOR= 3: SCALE= 1: ROT= 0
50 TP = 6.28318:K1 = 11811:K2 = 57.2958: REM

58 REM DEFINE ROUNDING FUNCTIONS
60 DEF FN R1(X) = INT (X * 10 + .5) / 10
61 DEF FN R2(X) = INT (X * 100 + .5) / 100
62 DEF FN R3(X) = INT (X * 1000 + .5) / 1000: REM

68 REM DISPLAY MENU FOR USER
69 PRINT "TYPE (H) TO HALT": HTAB 6: PRINT "(L) TO LIST MAIN PROGRAM":
  HTAB 6: PRINT "(R) TO RUN ANALYSIS": HTAB 6: PRINT "(S) TO LOAD FRESH
  SMITH CHART"
70 HTAB 6: PRINT "(T) TO MAKE HARD COPY OF DATA "
71 HTAB 6: PRINT "(U) TO LOAD UTILITIES FROM DISK": HTAB 6: PRINT
  "(P) TO PRINT GRAPHIC TABLE"
72 PRINT "(H),(L),(R),(S),(T),(U),OR (P)?": GET P$: PRINT CHR$(0):
  HOME
74 IF P$ = "H" THEN TEXT : HOME : END
75 IF P$ = "L" THEN GOTO 4200
76 IF P$ = "R" THEN GOTO 84
78 IF P$ = "S" THEN GOTO 5014
79 IF P$ = "T" THEN GOTO 9300
80 IF P$ = "U" THEN GOTO 5000
81 IF P$ = "P" THEN GOTO 6500
82 GOTO 68: REM DISPLAY MENU
84 REM BEGIN ANALYSIS
90 IZ = 0
95 I = PEEK (8125) + 1
96 POKE 8125,I: REM CHANGE PLOT SHAPE
98 UTAB (21): REM

```

```

99  REM  BEGIN DEFINITION OF PROBLEM
100  REM  SEVEN SECTION, CHEBYSCHEV, COAXIAL LPF.  0.1 DB RIPPLE.
110  R0 = 50: REM  DEFINE CENTER OF SMITH CHART
120  FOR F = 1.0 TO 15.0 STEP 1
130  W = TP * F:RN = 50.0:ZN = 0.0: REM  TERMINATION = 50 OHMS
150  Z0 = 15:LN = 27: GOSUB 1400: REM  1ST SECTION
170  Z0 = 85:LN = 170: GOSUB 1400: REM  2ND SECTION
190  Z0 = 15:LN = 49: GOSUB 1400: REM  3RD SECTION
210  Z0 = 85:LN = 194: GOSUB 1400: REM  4TH SECTION
230  Z0 = 15:LN = 49: GOSUB 1400: REM  5TH SECTION
250  Z0 = 85:LN = 170: GOSUB 1400: REM  6TH SECTION
270  Z0 = 15:LN = 27: GOSUB 1400: REM  7TH SECTION
290  GOSUB 9000: GOSUB 8000: REM  PLOT POINT ON SMITH CHART
300  NEXT F
999  GOTO 72: REM          END ANALYSIS, GO TO PROMPT

1000  REM  CONVERT IMPEDANCE TO ADMITTANCE
1001  X = L * W:RE = R
1005  IF C < > 0 THEN X = X - 1000 / (C * W)
1010  D = RE ^ 2 + X ^ 2:G = RE / D:B = - X / D: REM  INVERT Z=RE+JX
    TO Y=G+JB
1020  RETURN : REM

1100  REM  CONVERT ADMITTANCE TO IMPEDANCE
1101  G = 1 / R:B = C * W / 1000
1105  IF L < > 0 THEN B = B - 1 / (L * W)
1110  D = G ^ 2 + B ^ 2:RE = G / D:X = - B / D: REM  INVERT Y=G+JB
    TO Z=RE+JX
1120  RETURN : REM

1200  REM  SERIES ELEMENT: SERIES R,L,C
1201  X = L * W
1205  IF C < > 0 THEN X = X - 1000 / (C * W)
1210  RN = R + RN:ZN = X + ZN
1220  RETURN : REM

1300  REM  SHUNT ELEMENT,SERIES R,L,C
1304  RE = R:X = L * W
1305  IF C < > 0 THEN X = X - 1000 / (C * W)
1310  REM  SHUNT ELEMENT, RE+JX
1315  GOSUB 1010:G1 = G:B1 = B
1320  RE = RN:X = ZN: GOSUB 1010
1330  G = G1 + G:B = B1 + B
1340  GOSUB 1110:RN = RE:ZN = X
1350  RETURN : REM

1400  REM  SERIES TRANSMISSION LINE WITH CHARACTERISTIC IMPEDANCE Z0
1410  TN = TAN (W * LN / K1)
1420  AR = RN:AI = ZN + Z0 * TN
1430  BR = 1 - ZN * TN / Z0:BI = RN * TN / Z0
1440  GOSUB 2200:RN = CR:ZN = CI: REM  CALCULATE RN+J ZN=(AR+J AI)/(BR+J
    BI)
1450  RETURN : REM

1500  REM  ZN TO GAMMA (REFLECTION COEFFICIENT SR+J SI); ASSUMES REAL

```

```

R0; DEFAULT R0=50 MS
1505 IF R0 = 0 THEN R0 = 50.0
1510 AR = RN - R0:AI = XN
1520 BR = RN + R0:BI = XN
1530 GOSUB 2200:SR = CR:SI = CI: REM CALCULATE SR+J SI=(ZN-R0)/(ZN+R0)
1540 RETURN : REM

```

```

1600 REM GAMMA TO ZN; ASSUMES REAL R0; DEFAULT R0=50 OHMS
1605 IF R0 = 0 THEN R0 = 50.0
1610 AR = R0(1 + SR):AI = R0 * SI
1620 BR = 1 - SR:BI = - SI
1630 GOSUB 2200:RN = CR:XN = CI: REM CALCULATE RN+J XN=R0(1+SR+J
SI)/(1-SR-J SI)
1640 RETURN : REM

```

```

2000 REM COMPLEX MULTIPLY C=A*B
2010 CR = AR * BR - AI * BI
2020 CI = AR * BI + AI * BR
2030 RETURN : REM

```

```

2100 REM COMPLEX RECIPROCAL C=1/B
2110 AR = 1:AI = 0: REM

```

```

2200 REM COMPLEX DIVIDE, C=A/B
2210 D = BR ^ 2 + BI ^ 2
2220 CR = (AR * BR + AI * BI) / D
2230 CI = (AI * BR - AR * BI) / D
2240 RETURN : REM

```

```

4200 REM LIST MAIN PROGRAM
4210 TEXT : HOME : POKE 33,30: GOSUB 5300: REM TURN PRINTER ON
4220 POKE 1913,6: REM LEFT MARGIN
4240 LIST 100,999
4250 PRINT D$;"PR#0": REM TURN PRINTER OFF
4260 POKE 33,40: HOME
4270 GOTO 68: REM DISPLAY MENU

```

```

5000 REM LOADS SHAPES,HGR CHARACTERS,AND SMITH CHART
5001 D$ = CHR$(4)
5003 PRINT D$;"BLOAD SHAPES"
5004 POKE 232,191: POKE 233,31
5010 PRINT D$;"BLOAD HGR CHR GEN"
5014 REM LOAD SMITH CHART INTO GRAPHICS
5015 HGR
5020 PRINT D$;"BLOAD NUMB SMITH,A3192"
5025 POKE 8125,0: REM PLOT SHAPE PARAMETER IS IN 8125
5030 TEXT : HOME
5040 SCALE= 1: ROT= 0: HCOLOR= 3
5090 GOTO 68: REM DISPLAY MENU

```

```

5300 REM TURNS ON TRENDCOM PRINTER
5305 D$ = CHR$(4): REM D$=CTRL D
5307 PRINT CHR$(0)
5310 PRINT D$;"PR# 1": PRINT CHR$(0)

```

```

5320 POKE 1913,0: POKE 1785,70: REM MARGINS
5330 POKE 1657,80: REM LINE LENGTH
5340 RETURN : REM

```

```

6500 REM DUMPS GRAPHICS TABLE TO THE PRINTER, NO ROTATION, NORMAL
SIZE.
6510 POKE - 16301,0: POKE - 16300,0: POKE - 16297,0: POKE - 16304,0:
REM DISPLAY GRAPHICS TABLE
6520 VTAB 23: INPUT "LABEL?";LA$
6540 PRINT D$;"PR#1": PRINT TAB( 8);LA$: PRINT D$;"PR#0": REM
PRINT LABEL
6550 POKE 1784,15: POKE 1912,170: POKE 1272,0: POKE 1144,32: POKE
1528,0: POKE 1400,1: POKE 1656,0: REM SET PRINTER GRAPHICS TEMPORARIES
6560 CALL - 16000: REM CALL PRINTER SERVICE ROUTINE
6580 TEXT : HOME : GOTO 68: REM DISPLAY MENU

```

```

8000 REM SMITH PLOT
8002 ON I GOSUB 8006,8007,8008,8009
8003 GOTO 8010
8006 SH = 2: ROT= 0: SCALE= 1: RETURN
8007 SH = 2: ROT= 8: SCALE= 2: RETURN
8008 SH = 2: ROT= 0: SCALE= 2: RETURN
8009 SH = 2: ROT= 0: SCALE= 3: RETURN
8010 POKE - 16304,0: POKE - 16300,0
8011 POKE - 16297,0: POKE - 16301,0
8012 HCOLOR= 3
8020 RM = 79:XC = 140:YC = 80
8030 XP = XC + SR * RM
8040 YP = YC - SI * RM
8050 DRAW SH AT XP,YP
8060 RETURN : REM

```

```

8500 REM
8501 RETURN : REM

```

```

8510 CALL 25674: REM INIT HGR CHR GEN
8520 PRINT CHR$( 1); CHR$( 17)
8530 HCOLOR= 3
8535 DRAW SH AT 1,7 * I
8540 INPUT "COMMENT? ";CM$
8550 VTAB (I): HTAB (2): PRINT CM$
8560 CALL 1013: RETURN : REM

```

```

9000 REM PRINT ON CRT
9005 FZ = FN R2(F)
9010 RZ = FN R1(RN)
9020 XZ = FN R1(XN)
9022 GOSUB 1500
9024 SM = SQR (SR ^ 2 + SI ^ 2)
9026 SA = K2 * ATN (SI / SR)
9027 IF SR < 0 THEN SA = SA + 180
9028 IF SA > 180 THEN SA = SA - 360
9035 IF SM > = 1 THEN SWR = IN
9040 IF SM < 1 THEN SHR = (1 + SM) / (1 - SM)
9045 SM = FN R3(SM):SA = FN R1(SA):SWR = FN R2(SWR): REM ROUND
DATA FOR DISPLAY

```

```

9070 PRINT FZ; TAB( 8);RZ; TAB( 16);XZ; TAB( 24);SM; TAB( 32);SA
9080 GOSUB 9100: REM STORE DATA IN ARRAYS
9090 RETURN : REM
9100 REM STORES F, ZN, AND GAMMA IN ARRAYS
9110 IZ = IZ + 1
9120 FS(IZ) = FZ:RS(IZ) = RZ:XS(IZ) = XZ
9130 MS(IZ) = SM:AS(IZ) = SA:US(IZ) = SWR
9140 RETURN : REM

9200 REM PRINTS TABLE OF CALCULATED DATA
9201 HOME : GOSUB 5300: REM TURN ON PRINTER
9203 I$ = CHR$(9): REM I$=CTRL I
9205 PRINT I$;"N": REM DISABLE VIDEO
9207 POKE 33,80: REM SET APPLESOFT LINE LENGTH
9220 REM PRINT HEADING
9230 PRINT TAB( 8);"F"; TAB( 16);"REAL Z"; TAB( 24);"IMAG Z";
9240 PRINT TAB( 32);"/GAMMA/"; TAB( 40);"ANGLE"; TAB( 48);"USWR"
9245 FOR J = 1 TO 56: PRINT "-";: NEXT J
9250 PRINT ""
9255 REM PRINT DATA
9260 FOR J = 1 TO IZ
9275 PRINT TAB( 8);FS(J); TAB( 16);RS(J); TAB( 24);XS(J);
9280 PRINT TAB( 32);MS(J); TAB( 40);AS(J); TAB( 48);US(J)
9285 NEXT J
9290 PRINT D$;"PR#0": REM TURN PRINTER OFF
9292 POKE 33,40: PRINT I$;"I": REM RESTORE VIDEO
9295 TEXT : HOME : RETURN : REM

9300 REM DISPLAYS VIDEO TABLE OF CALCULATED DATA
9302 IF IZ = 0 THEN PRINT " ** NO TABLE **": PRINT "": GOTO 68
9305 J2% = 0: REM INITIALIZE ARRAY COUNTER
9310 TEXT : HOME
9320 REM PRINT HEADING
9330 PRINT "F"; TAB( 8);"REAL Z"; TAB( 16);"IMAG Z": PRINT ""
9340 J1% = 0: REM INITIALIZE SCREEN COUNTER
9350 J1% = J1% + 1:J2% = J2% + 1: PRINT FS(J2%); TAB( 8);RS(J2%); TAB(
  16);XS(J2%): REM PRINT LINE
9360 IF J2% = IZ THEN GOTO 9400: REM CHECK FOR END OF ARRAY
9370 IF J1% < 20 THEN GOTO 9350: REM CHECK FOR FULL SCREEN
9380 UTAB 23: PRINT "PRESS ANY KEY FOR REST OF TABLE";: GET P$: PRINT
  CHR$(0)
9390 GOTO 9310: REM PRINT REST OF ARRAY
9400 J2% = 0: REM INITIALIZE ARRAY COUNTER
9405 UTAB 23: PRINT "PRESS ANY KEY FOR REST OF TABLE";: GET P$: PRINT
  CHR$(0)
9410 TEXT : HOME
9420 REM PRINT HEADING
9430 PRINT "F"; TAB( 8);"/GAMMA/"; TAB( 16);"ANGLE"; TAB( 24);"USWR":
  PRINT ""
9440 J1% = 0: REM INITIALIZE SCREEN COUNTER
9450 J1% = J1% + 1:J2% = J2% + 1
9455 PRINT FS(J2%); TAB( 8);MS(J2%); TAB( 16);AS(J2%); TAB( 24);US(J2%):
  REM PRINT LINE
9460 IF J2% = IZ THEN GOTO 9500: REM CHECK FOR END OF ARRAY
9470 IF J1% < 20 THEN GOTO 9450: REM CHECK FOR FULL SCREEN
9480 UTAB 23: PRINT "PRESS ANY KEY FOR REST OF TABLE";: GET P$: PRINT
  CHR$(0)
9490 GOTO 9410: REM PRINT REST OF TABLE
9500 UTAB 23: PRINT "HARD COPY?(Y/N)";

9510 GET P$: PRINT CHR(0)
9520 IF P$ = "Y" THEN GOSUB 9200: TEXT : HOME : GOTO 68
9530 IF P$ = "N" THEN TEXT : HOME : GOTO 68
9540 GOTO 68

```



NATIONAL RADIO ASTRONOMY OBSERVATORY

ELECTRONICS DIVISION TECHNICAL NOTE NO. 96
ADDENDUM

TITLE: LADDER Analysis Program -- ADDENDUM

AUTHOR(S): Roger D. Norrod

DATE: January 4, 1982

Some revisions and additions have recently been made to the LADDER program at Green Bank. The revised program is stored on the NRAO Library 1.1 diskette under the name LADDER1. The changes are:

1. Three items have been added to the menu. These are:

- (M) TO DISPLAY MENU
- (G) TO DISPLAY GRAPHICS TABLE
- (Z) TO SAVE LINES 100-999 ON DISK

Item (Z) allows the user to save his analysis program and later EXEC it back into LADDER.

2. A soft reentry point has been provided. The user may restart LADDER after halting (with menu item H) by typing ESC 9.
3. Subroutine 1700 has been added. This routine combines the impedance $RN + j XN$ with the admittance of a shunt stub having impedance $Z\emptyset$ and length LN.
4. Subroutine 7000 has been added. This routine calculates and stores the reflection coefficient, VSWR, and reflection loss of the impedance $RN + j XN$ at frequency F. This could be used when studying filter networks and would be called instead of subroutine 9000.

A listing of the revised program follows.

Distribution:

M. Balister	J. Payne	R. Lacasse	TU Mountain File
S. Weinreb	P. Napier	H. Hvatum	TU Downtown File
GB Library	CV Library	VLA Library	C. Burgess

PROGRAM LENGTH= 9999 BYTES VARIABLES= 35 BYTES
 FREE MEMORY= 10835 BYTES
 START=16385 LOWEM=25994 FREE=26029 STRING=36864 HIGHEM=36864

```

10  REM          BLADDER VERSION 4JAN82

12  REM  MOVE PROGRAM TO HIGH MEMORY
14  PRINT CHR$(4);"BRUN LOWEM:": & LOWEM: 16384
16  HIGHEM: 36864
18  POKE 1013,76: POKE 1014,12: POKE 1015,151
20  POKE 1016,76: POKE 1017,0: POKE 1018,151
25  DIM U1(50),U2(50),U3(50),U4(50),U5(50),U6(50): REM

30  TEXT : HOME :U = 2
32  UTAB U: HTAB 16: PRINT "LADDER1"
33  UTAB U + 2: HTAB 8: PRINT "NETWORK ANALYSIS PROGRAM "
34  UTAB U + 4: HTAB 10: PRINT "NRAO--4JAN1982"
35  UTAB U + 6: PRINT "NORMAL UNITS ARE GHZ,MILS,NH, AND PF": PRINT
  "UTILITIES SHOULD BE LOADED AT START OF EACH SESSION.": UTAB U +
  10: REM
38  REM  INITIALIZE CONSTANTS
40  D$ = CHR$(4): REM  D$=CTRL D
45  IN = 1000
48  HCOLOR= 3: SCALE= 1: ROT= 0
50  TP = 6.28318:K1 = 11811:K2 = 57.2958: REM

52  REM  DEFINE ROUNDING FUNCTIONS
54  DEF FN R1(X) = INT (X * 10 + .5) / 10
56  DEF FN R2(X) = INT (X * 100 + .5) / 100
58  DEF FN R3(X) = INT (X * 1000 + .5) / 1000: REM

60  REM  DISPLAY MENU FOR USER
62  PRINT "TYPE (G) TO DISPLAY GRAPHICS": HTAB 6: PRINT "(H) TO HALT
  (USE ESC9 TO RESTART)": HTAB 6: PRINT "(L) FOR HARDCOPY LINES 100,999"
63  HTAB 6: PRINT "(M) TO DISPLAY MENU": HTAB 6: PRINT "(R) TO RUN
  PROGRAM 100": HTAB 6: PRINT "(S) TO LOAD FRESH SMITH CHART"
64  HTAB 6: PRINT "(T) TO DISPLAY TABLE OF DATA"
66  HTAB 6: PRINT "(U) TO LOAD UTILITIES FROM DISK": HTAB 6: PRINT
  "(P) TO PRINT GRAPHIC TABLE": HTAB 6: PRINT "(Z) TO SAVE MAIN PROGRAM"
68  PRINT : PRINT "ENTER SELECTION ": GET P$: PRINT CHR$(0): HOME

70  IF P$ = "H" THEN TEXT : HOME : END
71  IF P$ = "M" THEN TEXT : HOME : GOTO 60
72  IF P$ = "L" THEN GOTO 4200
73  IF P$ = "G" THEN GOTO 8500
74  IF P$ = "R" THEN GOTO 90
76  IF P$ = "S" THEN GOTO 5014
78  IF P$ = "T" THEN GOTO 9300
80  IF P$ = "U" THEN GOTO 5000
81  IF P$ = "P" THEN GOTO 6500
82  IF P$ = "Z" THEN GOTO 2500
89  GOTO 60: REM  DISPLAY MENU

90  REM  BEGIN ANALYSIS

```

```

92 IZ = 0
95 I = PEEK (8125) + 1
96 POKE 8125,I: REM    CHANGE PLOT SHAPE
98 UTAB (21): REM

```

```

99 REM BEGIN DEFINITION OF PROBLEM

```

INSERT USER PROGRAM HERE

```

999 GOTO 68: REM          END ANALYSIS DISPLAY PROMPT
1000 REM CONVERT IMPEDANCE TO ADMITTANCE
1001 X = L * W:RE = R
1005 IF C < > 0 THEN X = X - 1000 / (C * W)
1010 D = RE ^ 2 + X ^ 2:G = RE / D:B = - X / D: REM    INVERT Z=RE+JX
    TO Y=G+JB
1020 RETURN : REM

```

```

1100 REM CONVERT ADMITTANCE TO IMPEDANCE
1101 G = 1 / R:B = C * W / 1000
1105 IF L < > 0 THEN B = B - 1 / (L * W)
1110 D = G ^ 2 + B ^ 2:RE = G / D:X = - B / D: REM    INVERT Y=G+JB
    TO Z=RE+JX
1120 RETURN : REM

```

```

1200 REM SERIES ELEMENT: SERIES R,L,C
1201 X = L * W
1205 IF C < > 0 THEN X = X - 1000 / (C * W)
1210 RN = R + RN:YN = X + XN
1220 RETURN : REM

```

```

1300 REM SHUNT ELEMENT,SERIES R,L,C
1304 RE = R:X = L * W
1305 IF C < > 0 THEN X = X - 1000 / (C * W)
1310 REM SHUNT ELEMENT, RE+JX
1315 GOSUB 1010:G1 = G:B1 = B
1320 RE = RN:X = XN: GOSUB 1010
1330 G = G1 + G:B = B1 + B
1340 GOSUB 1110:RN = RE:XN = X
1350 RETURN : REM

```

```

1400 REM SERIES TRANSMISSION LINE WITH CHARACTERISTIC IMPEDANCE Z0
1410 TN = TAN (W * LN / K1)
1420 AR = RN:AI = XN + Z0 * TN
1430 BR = 1 - XN * TN / Z0:BI = RN * TN / Z0

```

BI)

1450 RETURN : REM

1500 REM ZN TO GAMMA (REFLECTION COEFFICIENT SR+J SI); ASSUMES REAL
R0; DEFAULT R0=50 OHMS
1505 IF R0 = 0 THEN R0 = 50.0
1510 AR = RN - R0:AI = XN
1520 BR = RN + R0:BI = XN
1530 GOSUB 2200:SR = CR:SI = CI: REM CALCULATE SR+J SI=(ZN-R0)/(ZN+R0)
1540 SM = SQR (SR ^ 2 + SI ^ 2)
1545 SA = K2 * ATN (SI / SR)
1550 IF SR < 0 THEN SA = SA + 180
1555 IF SA > 180 THEN SA = SA - 360
1560 IF SM > = .999 THEN SWR = IN
1565 IF SM < .999 THEN SWR = (1 + SM) / (1 - SM)
1570 RETURN : REM

1600 REM GAMMA TO ZN; ASSUMES REAL R0; DEFAULT R0=50 OHMS
1605 IF R0 = 0 THEN R0 = 50.0
1610 AR = R0(1 + SR):AI = R0 * SI
1620 BR = 1 - SR:BI = - SI
1630 GOSUB 2200:RN = CR:XN = CI: REM CALCULATE RN+J XN=R0(1+SR+J
SI)/(1-SR-J SI)
1640 RETURN : REM

1700 REM SHUNT SHORTED STUB, LENGTH LN, IMPEDANCE Z0
1710 RE = RN:X = XN: GOSUB 1010: REM CALC INPUT ADMITTANCE AND SAVE
1720 G1 = G:B1 = B
1730 RN = 0:XN = 0: GOSUB 1400: REM CALC INPUT IMPEDANCE OF STUB
1740 RE = RN:X = XN: GOSUB 1010: REM CALC STUB ADMITTANCE
1750 G = G + G1:B = B + B1: GOSUB 1110: REM COMBINE ADMITTANCES AND
CONVERT TO IMPEDANCE
1760 RN = RE:XN = X
1770 RETURN : REM
1900 RUN 25: REM REENTRY VIA ESC 9

2000 REM COMPLEX MULTIPLY C=A*B
2010 CR = AR * BR - AI * BI
2020 CI = AR * BI + AI * BR
2030 RETURN : REM

2100 REM COMPLEX RECIPROCAL C=1/B
2110 AR = 1:AI = 0: REM

2200 REM COMPLEX DIVIDE, C=A/B
2210 D = BR ^ 2 + BI ^ 2
2220 CR = (AR * BR + AI * BI) / D
2230 CI = (AI * BR - AR * BI) / D
2240 RETURN : REM

2500 REM CAPTURE LISTING
2505 TEXT : HOME

```

2514 PRINT ":"
2516 PRINT "THE TEXT FILE CAN BE RECALLED INTO"
2518 PRINT "THE LADDER1 PROGRAM BY:"
2520 PRINT TAB( 5);"FIRST - LOAD LADDER1"
2522 PRINT TAB( 4);"SECOND - EXEC FILENAME"
2524 PRINT TAB( 5);"THIRD - RUN"
2526 PRINT ":"
2530 INPUT "NOW, WHAT NAME WOULD YOU LIKE LINE NUMBERS 100-999 SAVED
UNDER? ";TF$
2535 PRINT " "; PRINT "THANK YOU, PLEASE WAIT"
2540 PRINT D$;"OPEN";TF$
2545 PRINT D$;"DELETE";TF$
2550 PRINT D$;"OPEN";TF$
2555 PRINT D$;"WRITE";TF$
2560 POKE 33,30
2565 LIST 100,999
2570 PRINT D$;"CLOSE";TF$
2575 TEXT : HOME : PRINT "LINES 100-999 SAVED UNDER FILENAME ";TF$
2580 UTAB 5: GOTO 60: REM DISPLAY MENU
4200 REM LIST MAIN PROGRAM
4210 TEXT : HOME : POKE 33,30: GOSUB 5300: REM TURN PRINTER ON
4220 POKE 1913,6: REM LEFT MARGIN
4240 LIST 100,999
4250 PRINT D$;"PR#0": REM TURN PRINTER OFF
4260 POKE 33,40: HOME
4270 GOTO 60: REM DISPLAY MENU

5000 REM LOADS SHAPES,HGR CHARACTERS,AND SMITH CHART
5001 D$ = CHR$( 4)
5003 PRINT D$;"BLOAD SHAPES,A8127": REM LENGTH 24 BYTES
5004 POKE 232,191: POKE 233,31
5010 PRINT D$;"BLOAD HGR CHR GEN,A2048": REM LENGTH 3328 BYTES
5014 REM LOAD SMITH CHART INTO GRAPHICS
5015 HGR
5020 PRINT D$;"BLOAD NUMB SMITH,A8192": REM LENGTH 8192 BYTES
5025 POKE 8125,0: REM PLOT SHAPE PARAMETER IS IN 8125
5030 TEXT : HOME
5040 SCALE= 1: ROT= 0: HCOLOR= 3
5090 GOTO 60: REM DISPLAY MENU
5300 REM TURNS ON TRENDCOM PRINTER
5305 D$ = CHR$( 4): REM D$=CTRL D
5307 PRINT CHR$( 0)
5310 PRINT D$;"PR# 1": PRINT CHR$( 0)
5320 POKE 1913,0: POKE 1785,70: REM MARGINS
5330 POKE 1657,80: REM LINE LENGTH
5340 RETURN : REM

6500 REM DUMPS GRAPHICS TABLE TO THE PRINTER, NO ROTATION, NORMAL
SIZE.
6510 POKE - 16301,0: POKE - 16300,0: POKE - 16297,0: POKE - 16304,0:
REM DISPLAY GRAPHICS TABLE
6520 UTAB 23: INPUT "LABEL?";LA$
6540 PRINT D$;"PR#1": PRINT TAB( 8);LA$: PRINT D$;"PR#0": REM
PRINT LABEL
6550 POKE 1784,15: POKE 1912,170: POKE 1272,0: POKE 1144,32: POKE
1528,0: POKE 1400,1: POKE 1656,0: REM SET PRINTER GRAPHICS TEMPORARIES
6560 CALL - 16000: REM CALL PRINTER SERVICE ROUTINE
6580 TEXT : HOME : GOTO 60: REM DISPLAY MENU

```

```

7000 REM PRINTS REFL COEFF, SWR, AND LOSS ON CRT
7010 GOSUB 1500: REM CALCULATE GAMMA AND USWR
7020 LOSS = - 4.342945 * LOG (1 - SM * SM): REM CALCULATE INSERTION
LOSS IN DB
7030 FZ = FN R2(F):LOSS = FN R2(LOSS):SWR = FN R2(SWR): REM ROUND
DATA
7040 SM = FN R3(SM):SA = FN R1(SA)
7050 PRINT FZ; TAB( 8);LOSS; TAB( 16);SWR; TAB( 24);SM; TAB( 32);SA
7055 U1$ = "FREQ":U2$ = "LOSS":U3$ = "USWR":U4$ = "/GAMMA/":U5$ = "ANGLE":
U6$ = ""
7060 GOSUB 7100: REM STORE DATA IN ARRAY
7070 RETURN : REM

```

```

7100 REM STORES F, LOSS, GAMMA, AND SWR IN ARRAYS
7110 IZ = IZ + 1
7120 U1(IZ) = FZ:U2(IZ) = LOSS:U3(IZ) = SWR:U4(IZ) = SM:U5(IZ) = SA
7140 RETURN : REM

```

```

8000 REM SMITH PLOT
8002 ON I GOSUB 8006,8007,8008,8009
8003 GOTO 8010
8006 SH = 2: ROT= 0: SCALE= 1: RETURN
8007 SH = 2: ROT= 8: SCALE= 2: RETURN
8008 SH = 2: ROT= 0: SCALE= 2: RETURN
8009 SH = 2: ROT= 0: SCALE= 3: RETURN
8010 POKE - 16304,0: POKE - 16300,0
8011 POKE - 16297,0: POKE - 16301,0
8012 HCOLOR= 3
8020 RM = 79:XC = 140:YC = 80
8030 XP = XC + SR * RM
8040 YP = YC - SI * RM
8050 DRAW SH AT XP,YP
8060 RETURN : REM

```

```

8500 REM DISPLAYS GRAPHICS TABLE AND
8510 REM THEN RETURNS TO PROMPT.
8520 POKE - 16304,0: POKE - 16300,0
8530 POKE - 16297,0: POKE - 16301,0
8540 UTAB 22: GOTO 68: REM

```

```

9000 REM PRINT ON CRT
9005 FZ = FN R2(F)
9010 RZ = FN R3(RN)
9020 XZ = FN R3(XN)
9022 GOSUB 1500
9045 SM = FN R3(SM):SA = FN R1(SA):SWR = FN R2(SWR): REM ROUND
DATA FOR DISPLAY
9070 PRINT FZ; TAB( 8);RZ; TAB( 16);XZ; TAB( 24);SM; TAB( 32);SA
9075 U1$ = "FREQ":U2$ = "REAL Z":U3$ = "IMAG Z":U4$ = "/GAMMA/":U5$
= "ANGLE":U6$ = "USWR"
9080 GOSUB 9100: REM STORE DATA IN ARRAYS
9090 RETURN : REM
9100 REM STORES F, ZN, AND GAMMA IN ARRAYS
9110 IZ = IZ + 1

```

```
9140 RETURN : REM
```

```
9200 REM PRINTS TABLE OF CALCULATED DATA
9201 HOME : GOSUB 5300: REM TURN ON PRINTER
9203 I$ = CHR$(9): REM I$=CTRL I
9205 PRINT I$;"N": REM DISABLE VIDEO
9207 POKE 33,80: REM SET APPLESOFT LINE LENGTH
9220 REM PRINT HEADING
9230 PRINT TAB( 8);U1$; TAB( 16);U2$; TAB( 24);U3$;
9240 PRINT TAB( 32);U4$; TAB( 40);U5$; TAB( 48);U6$
9245 FOR J = 1 TO 56: PRINT "-";: NEXT J
9250 PRINT ""
9255 REM PRINT DATA
9260 FOR J = 1 TO IZ
9275 PRINT TAB( 8);U1(J); TAB( 16);U2(J); TAB( 24);U3(J);
9280 PRINT TAB( 32);U4(J); TAB( 40);U5(J); TAB( 48);U6(J)
9285 NEXT J
9290 PRINT 0$;"PR#0": REM TURN PRINTER OFF
9292 POKE 33,40: PRINT I$;"I": REM RESTORE VIDEO
9295 TEXT : HOME : RETURN : REM
```

```
9300 REM DISPLAYS VIDEO TABLE OF CALCULATED DATA
9302 IF IZ = 0 THEN PRINT " ** NO TABLE **": PRINT "": GOTO 68
9305 J2% = 0: REM INITIALIZE ARRAY COUNTER
9310 TEXT : HOME
9320 REM PRINT HEADING
9330 PRINT U1$; TAB( 8);U2$; TAB( 16);U3$: PRINT ""
9340 J1% = 0: REM INITIALIZE SCREEN COUNTER
9350 J1% = J1% + 1: J2% = J2% + 1: PRINT U1(J2%); TAB( 8);U2(J2%); TAB(
  16);U3(J2%): REM PRINT LINE
9360 IF J2% = IZ THEN GOTO 9400: REM CHECK FOR END OF ARRAY
9370 IF J1% < 20 THEN GOTO 9350: REM CHECK FOR FULL SCREEN
9380 UTAB 23: PRINT "PRESS ANY KEY FOR REST OF TABLE";: GET P$: PRINT
  CHR$(0)
9390 GOTO 9310: REM PRINT REST OF ARRAY
9400 J2% = 0: REM INITIALIZE ARRAY COUNTER
9405 UTAB 23: PRINT "PRESS ANY KEY FOR REST OF TABLE";: GET P$: PRINT
  CHR$(0)
9410 TEXT : HOME
9420 REM PRINT HEADING
9430 PRINT U1$; TAB( 8);U4$; TAB( 16);U5$; TAB( 24);U6$: PRINT ""
9440 J1% = 0: REM INITIALIZE SCREEN COUNTER
9450 J1% = J1% + 1: J2% = J2% + 1
9455 PRINT U1(J2%); TAB( 8);U4(J2%); TAB( 16);U5(J2%); TAB( 24);U6(J2%):
  REM PRINT LINE
9460 IF J2% = IZ THEN GOTO 9500: REM CHECK FOR END OF ARRAY
9470 IF J1% < 20 THEN GOTO 9450: REM CHECK FOR FULL SCREEN
9480 UTAB 23: PRINT "PRESS ANY KEY FOR REST OF TABLE";: GET P$: PRINT
  CHR$(0)
9490 GOTO 9410: REM PRINT REST OF TABLE
9500 UTAB 23: PRINT "HARD COPY?(Y/N)";
9510 GET P$: PRINT CHR(0)
9520 IF P$ = "Y" THEN GOSUB 9200: TEXT : HOME : GOTO 60
9530 IF P$ = "N" THEN TEXT : HOME : GOTO 60
9540 GOTO 9500: REM
```